1 Optical Fibers — supplementary notes

Optical fibers are cylindrical dielectric waveguides. Their operation, in analogy with dielectric slab waveguides, depends on total internal reflections from the boundary of a high refractive index cylindrical *core* embedded in a lower refractive index *cladding.* They support TE and TM modes just like slab waveguides, but these modes are not "fundamental" within the optical fiber — that is, they have finite cutoff frequencies greater than a cutoff frequency of yet another mode, the fundamental one, that happens to be, in step-index optical fibers (as opposed to graded-index), one particular one of an infinitely many *hybrid modes* having non-zero E_z *and* H_z variations. To understand the total collection of TE, TM, and hybrid HE and EH modes that populate our step-index optical fibers we need to look at the solutions of Maxwell's wave equation in a cylindrical geometry.

1.1 Wave equation in cylindrical coordinates

Maxwell's curl equations furnish us with vector wave equations having the z-components

$$\begin{aligned} \nabla^2 E_z + k^2 E_z &= 0 \\ \nabla^2 H_z + k^2 H_z &= 0 \end{aligned} \tag{1}$$

where

$$k^2 = \frac{\omega^2}{c^2} n^2 = k_o^2 n^2 \tag{2}$$

with $n = \sqrt{\mu_r \epsilon_r}$ denoting the refractive index in a homogeneous medium[1]. In a cylindrical geometry of a *step-index fiber* with (r, ϕ, z) coordinates we seek propagating solutions of a separable form

$$E_z, H_z \propto R(r) P(\phi) e^{-j\beta z}, \tag{3}$$

in which case the equations above reduce to

$$\nabla_\perp^2 RP + (k^2 - \beta^2) RP = \frac{1}{r}\frac{\partial}{\partial r}(r \frac{\partial PR}{\partial r}) + \frac{1}{r^2}\frac{\partial^2}{\partial \phi^2}(PR) + (k^2 - \beta^2) RP = 0$$

that further simplifies as

$$\frac{r}{R}\frac{\partial}{\partial r}(r\frac{\partial R}{\partial r}) + \frac{1}{P}\frac{\partial^2 P}{\partial \phi^2} + r^2(k^2 - \beta^2) = 0.$$

Let

$$\frac{1}{P}\frac{\partial^2 P}{\partial \phi^2} = -m^2 \quad \Rightarrow \quad \frac{\partial^2 P}{\partial \phi^2} + m^2 P = 0 \quad \Rightarrow \quad P(\phi) \propto e^{\pm jm\phi} \tag{4}$$

and consequently

$$\frac{r}{R}\frac{\partial}{\partial r}(r\frac{\partial R}{\partial r}) + r^2(k^2 - \beta^2) = m^2 \quad \Rightarrow \quad R(r) \propto J_m(\sqrt{k^2 - \beta^2}r),\, K_m(\sqrt{\beta^2 - k^2}r). \tag{5}$$

[1] Also valid in inhomogeneous graded-index fibers with an index profile $n(r) = n_1 - f(r)\delta n$, with $0 = f(0) < f(r) < 1 = f(a)$, so long as "weakly guiding" condition $\delta n \ll n_1$ holds. Since the *Helmholtz* equation (1) with inhomogeneous $k(r)$ is also the *Klein-Gordon* equation describing the quantum wavefunctions $\psi(\mathbf{r})$ of massless spin-zero particles confined in potential wells $\propto f(r)$, graded-index fibers are found to support quantized $E_z(r)$ variations (modes) matching the $\psi(r)$ eigenfunctions of quantized energy states of such particles in cylindrical potential wells. See *Gloge* [Rep. Prog. Phys., **42**, 1979] for a discussion of $E_z(r)$ "eigenfunctions" resulting from a parabolic $f(r) = r^2/a^2$ profile, analogous to the quantum mechanical harmonic oscillator and leading to a ladder of β^2/k_o^2 "eigenvalues" varying with a pair of "quantum numbers" m and l (instead of 3 as in a spherical hydrogen atom) for a total of $V^2/4$ modes expressed in terms of the *V-number* (11) of the fiber. Multimode graded-index fibers exhibit weaker "modal dispersion" — due to different group velocities — than multimode step-index fibers because of a compensation effect involving path lengths and propagation speeds.

Single valued $P(\phi)$ demands integer values of m above, while $J_m(X)$ and $K_m(Y)$ denote mth order Bessel functions of the 1st kind and modified 2nd kind, respectively, fitting the linear differential equation in the left known as Bessel's differential equation. We disregard other solutions — 2nd kind and modified 1st kind — of Bessel's differential equation leading to unbounded fields in core and cladding regions for reasons of realizability.

With the above 1D solutions we can construct bounded 3D phasor solutions of the wave equation as

$$\begin{aligned} E_z &= J_m(k_t r)(Ae^{jm\phi} + cc)e^{-j\beta z} \\ H_z &= J_m(k_t r)(Be^{jm\phi} + cc)e^{-j\beta z} \end{aligned} \tag{6}$$

in the core region $r < a$ where $n = n_1$ and

$$k_t^2 \equiv k^2 - \beta^2 = k_o^2 n_1^2 - \beta^2. \tag{7}$$

Likewise, for the cladding region where $n = n_2 < n_1$ we use the bounded options

$$\begin{aligned} E_z &= K_m(\alpha r)(Ce^{jm\phi} + cc)e^{-j\beta z} \\ H_z &= K_m(\alpha r)(De^{jm\phi} + cc)e^{-j\beta z} \end{aligned} \tag{8}$$

where

$$\alpha^2 \equiv \beta^2 - k^2 = \beta^2 - k_o^2 n_2^2. \tag{9}$$



(a) (b)

Figure 1: (a) The red curve depicts $J_1(r)$ in the range $0 < r < 3$. The green curve is $K_1(r)$ normalized by $K_1(3)/J_1(3)$ plotted over $2 < r < 10$. Their combination illustrates how J_m and K_m functions can be patched together at some $r = a$ boundary to construct bounded field profiles for optical fiber modes. (b) Plots of $J_m(r)$ and $K_m(r)$ for $m = 0, 1, 2$.

Note that

$$\alpha^2 + k_t^2 = k_o^2(n_1^2 - n_2^2) \equiv \frac{V^2}{a^2}. \tag{10}$$

This leads to

$$V = k_o a\sqrt{n_1^2 - n_2^2} = \frac{\omega}{c} a\text{N.A.} \tag{11}$$

which is a normalized operation frequency of the fiber known as its *V-number,* while

$$\text{N.A.} \equiv \sqrt{n_1^2 - n_2^2} = \sin\theta_a \tag{12}$$

is the so-called *numerical aperture* of the fiber corresponding to the sine of the acceptance angle — rays making incidence angle $\theta < \theta_a$ with the normal to the external cross-section of the fiber can be guided by the fiber since the internal incidence angle at the core-cladding boundary will then exceed the the critical angle $\theta_c = \sin^{-2} n_2/n_1$ as required for total internal reflection (HW problem).

Given E_z, H_z, the remaining field components in the transverse plane can be obtained from Maxwell's curl equations as follows (HW problem):

$$\begin{aligned} E_\phi &= \frac{-j}{k^2-\beta^2}\left(\frac{\beta}{r}\frac{\partial E_z}{\partial \phi} - \omega\mu\frac{\partial H_z}{\partial r}\right) \\ E_r &= \frac{-j}{k^2-\beta^2}\left(\frac{\omega\mu}{r}\frac{\partial H_z}{\partial \phi} + \beta\frac{\partial E_z}{\partial r}\right) \\ H_\phi &= \frac{-j}{k^2-\beta^2}\left(\omega\epsilon\frac{\partial E_z}{\partial r} + \frac{\beta}{r}\frac{\partial H_z}{\partial \phi}\right) \\ H_r &= \frac{-j}{k^2-\beta^2}\left(\beta\frac{\partial H_z}{\partial r} - \frac{\omega\epsilon}{r}\frac{\partial E_z}{\partial \phi}\right). \end{aligned} \tag{13}$$

To determine the amplitude coefficients above, and the propagation constant β, we need to enforce the boundary conditions at $r = a$ —the core radius of the fiber — of the continuity of tangential E_z, E_ϕ as well as H_z, H_ϕ.

Before doing that, we note that one of the amplitude coefficients above, say A, can be selected at will. Then the boundary conditions will identify the remaining coefficients in terms of A. Let us briefly see the impact of choosing $A = \frac{1}{2}$ and $A = \frac{1}{2j}$ as two plausible examples.

Illustrative example: With $A = \frac{1}{2}$ we get

$$E_z = J_m(k_t r)(Ae^{jm\phi} + cc)e^{-j\beta z} = J_m(k_t r)\cos(m\phi)e^{-j\beta z} \tag{14}$$

while with $A = \frac{1}{2j}$

$$E_z = J_m(k_t r)(Ae^{jm\phi} + cc)e^{-j\beta z} = J_m(k_t r)\sin(m\phi)e^{-j\beta z} \tag{15}$$

- Now, by convention, $\phi = 0$ is the direction of the x-axis while $\phi = 90°$ is the direction of y-axis.
  - Accordingly, these two solutions above represent having the E_z field intensity peaking along the x- and y-axes, respectively (for $m = 1$ case, at least). When we want to have the field intensity peaking along some other direction we will need to use a linear combination of these two solutions.
- Hence it is natural to think of these two solutions as a c*omplete set of orthogonal basis functions* or *eigenfunctions* for E_z. In general, for any $m \geq 1$, these will have a $90°/m$ rotational symmetry property — rotating the intensity pattern of any one of these eigenfunctions by $90°/m$ in the xy-plane will produce the intensity pattern of the complementary function.
- The case $m = 0$ is special — in that case there is no ϕ variation of field intensity and $\sin(m\phi)$ basis function does not exist!
- We will refer to the the orthogonal eigensolution pairs described here as "normal modes" in the following discussion.

By applying the boundary conditions above it can be shown that $B = j\xi A$, where ξ is a real number. This leads to

$$H_z \propto \left\{ \begin{array}{c} \sin(m\phi) \\ \cos(m\phi) \end{array} \right\} \text{ when } E_z \propto \left\{ \begin{array}{c} \cos(m\phi) \\ \sin(m\phi) \end{array} \right\}.$$

Accordingly it is more convenient to carry out the boundary condition matching exercise with a modified set of candidate solutions — normal modes — written as

$$\begin{aligned} E_z &= AJ_m(k_t r)\cos(m\phi)e^{-j\beta z} \\ H_z &= BJ_m(k_t r)\sin(m\phi)e^{-j\beta z} \end{aligned} \tag{16}$$

and

$$\begin{aligned} E_z &= CK_m(\alpha r)\cos(m\phi)e^{-j\beta z} \\ H_z &= DK_m(\alpha r)\sin(m\phi)e^{-j\beta z} \end{aligned} \tag{17}$$

and obtain the complementary partner after applying a $90°/m$ rotation in the solution obtained for the first basis function.

1.2 Boundary condition matching

Using the candidate solutions just proposed, and the chain rule of differentiation, the transverse field components are found to be in the core region

$$\begin{aligned} E_\phi &= \frac{-j}{k^2-\beta^2}\left(\frac{\beta}{r}\frac{\partial E_z}{\partial \phi} - \omega\mu\frac{\partial H_z}{\partial r}\right) = \frac{-j}{k_t^2}\left(\frac{-m\beta}{r}AJ_m(k_t r)\sin(m\phi) - \omega\mu_1 B k_t J_m'(k_t r)\sin(m\phi)\right)e^{-j\beta z} \\ E_r &= \frac{-j}{k^2-\beta^2}\left(\frac{\omega\mu}{r}\frac{\partial H_z}{\partial \phi} + \beta\frac{\partial E_z}{\partial r}\right) = \frac{-j}{k_t^2}\left(\frac{m\omega\mu_1}{r}BJ_m(k_t r)\cos(m\phi) + \beta k_t A J_m'(k_t r)\cos(m\phi)\right)e^{-j\beta z} \\ H_\phi &= \frac{-j}{k^2-\beta^2}\left(\omega\epsilon\frac{\partial E_z}{\partial r} + \frac{\beta}{r}\frac{\partial H_z}{\partial \phi}\right) = \frac{-j}{k_t^2}\left(\omega\epsilon_1 k_t A J_m'(k_t r)\cos(m\phi) + \frac{m\beta}{r}BJ_m(k_t r)\cos(m\phi)\right)e^{-j\beta z} \\ H_r &= \frac{-j}{k^2-\beta^2}\left(\beta\frac{\partial H_z}{\partial r} - \frac{\omega\epsilon}{r}\frac{\partial E_z}{\partial \phi}\right) = \frac{-j}{k_t^2}\left(\beta k_t B J_m'(k_t r)\sin(m\phi) + \frac{m\omega\epsilon_1}{r}AJ_m(k_t r)\sin(m\phi)\right)e^{-j\beta z} \end{aligned} \tag{18}$$

and in the cladding region

$$\begin{aligned} E_\phi &= \frac{-j}{k^2-\beta^2}\left(\frac{\beta}{r}\frac{\partial E_z}{\partial \phi} - \omega\mu\frac{\partial H_z}{\partial r}\right) = \frac{j}{\alpha^2}\left(\frac{-m\beta}{r}CK_m(\alpha r)\sin(m\phi) - \omega\mu_2 D\alpha K_m'(\alpha r)\sin(m\phi)\right)e^{-j\beta z} \\ E_r &= \frac{-j}{k^2-\beta^2}\left(\frac{\omega\mu}{r}\frac{\partial H_z}{\partial \phi} + \beta\frac{\partial E_z}{\partial r}\right) = \text{HW problem} \\ H_\phi &= \frac{-j}{k^2-\beta^2}\left(\omega\epsilon\frac{\partial E_z}{\partial r} + \frac{\beta}{r}\frac{\partial H_z}{\partial \phi}\right) = \frac{j}{\alpha^2}\left(\omega\epsilon_2\alpha C K_m'(\alpha r)\cos(m\phi) + \frac{m\beta}{r}DK_m(\alpha r)\cos(m\phi)\right)e^{-j\beta z} \\ H_r &= \frac{-j}{k^2-\beta^2}\left(\beta\frac{\partial H_z}{\partial r} - \frac{\omega\epsilon}{r}\frac{\partial E_z}{\partial \phi}\right) = \text{HW problem.} \end{aligned} \tag{19}$$

Our four boundary conditions subsequently yield, with

$$X \equiv k_t a \text{ and } Y \equiv \alpha a \text{ and } V = \sqrt{X^2+Y^2}, \tag{20}$$

the following four relations:

$$\begin{aligned} AJ_m(X) &= CK_m(Y) \quad \text{—}E_z\text{ matching} \\ BJ_m(X) &= DK_m(Y) \quad \text{—}H_z\text{ matching} \\ \frac{1}{X^2}(m\beta A J_m(X) + \omega\mu_1 B X J_m'(X)) &= \frac{-1}{Y^2}(m\beta C K_m(Y) + \omega\mu_2 D Y K_m'(Y)) \quad \text{—}E_\phi\text{ matching} \\ \frac{-1}{X^2}(\omega\epsilon_1 X A J_m'(X) + m\beta B J_m(X)) &= \frac{1}{Y^2}(\omega\epsilon_2 Y C K_m'(Y) + m\beta D K_m(Y)) \quad \text{—}H_\phi\text{ matching} \end{aligned} \tag{21}$$

where $J_m'(X)$ and $K_m'(Y)$ indicate Bessel function derivatives — see below. Note that the cancellation of ϕ dependent terms from these equations depended crucially on having normal mode $E_z \propto \cos(m\phi)$ paired with $H_z \propto \sin(m\phi)$ — this result can therefore be taken as *a posteriori* justification of the form assumed for our normal modes.

Now, using the first two equations to get rid of C and D in favor of A and B, we have

$$\begin{aligned}\frac{1}{X^2}(m\beta A J_m(X) + \omega\mu_1 B X J_m'(X)) &= \frac{-1}{Y^2}(m\beta A J_m(X) + \omega\mu_2 B \frac{J_m(X)}{K_m(Y)} Y K_m'(Y)) \\ \frac{-1}{X^2}(\omega\epsilon_1 X A J_m'(X) + m\beta B J_m(X)) &= \frac{1}{Y^2}(\omega\epsilon_2 Y A \frac{J_m(X)}{K_m(Y)} K_m'(Y) + m\beta B J_m(X)),\end{aligned}$$

yielding, after some more work,

$$B(\frac{\mu_{1r} J_m'(X)}{X J_m(X)} + \frac{\mu_{2r} K_m'(Y)}{Y K_m(Y)}) = -A(\frac{1}{X^2} + \frac{1}{Y^2})\frac{m\beta}{\omega\mu_o} \tag{22}$$

$$A(\frac{\epsilon_{1r} J_m'(X)}{X J_m(X)} + \frac{\epsilon_{2r} K_m'(Y)}{Y K_m(Y)}) = -B(\frac{1}{X^2} + \frac{1}{Y^2})\frac{m\beta}{\omega\epsilon_o}. \tag{23}$$

Now, multiply the equations and simplify using $\mu_{2r} = \mu_{1r} = 1$ to obtain the dispersion relation

$$(\frac{J_m'(X)}{X J_m(X)} + \frac{K_m'(Y)}{Y K_m(Y)})(\frac{n_1^2 J_m'(X)}{X J_m(X)} + \frac{n_2^2 K_m'(Y)}{Y K_m(Y)}) = \frac{m^2\beta^2}{k_o^2}(\frac{1}{X^2} + \frac{1}{Y^2})^2. \tag{24}$$

We call (24) the *dispersion relation* because given the frequency $\omega = k_o c$, it can be solved for the propagation constant β, once the fiber parameters a, n_1,and n_2 are specified.

Recall that $X = ak_t = a\sqrt{k_o^2 n_1^2 - \beta^2}$ and $Y = \sqrt{V^2 - X^2}$ where $V = k_o a\sqrt{n_1^2 - n_2^2}$.

In computing the dispersion relation (24) we will typically plot its RHS and LHS as a function of X for a given V and compute k_t on the way to β from the X-numbers at the intersections of the RHS and LHS curves. In evaluating dispersion relation (24) use the Bessel identities

$$J_m'(X) = \mp J_{m\pm1}(X) \pm \frac{m}{X} J_m(X) \tag{25}$$

and

$$(-1)^m K_m'(Y) = (-1)^{m\pm1} K_{m\pm1}(Y) \pm \frac{m}{Y}(-1)^m K_m(Y). \tag{26}$$

Finally, we note that dispersion relation (24) is equally valid for $E_z \propto \sin(m\phi)$ modes, $m \geq 0$, although the plots of field intensities that can be constructed using the equations derived above will need to be rotated by $90°/m$ to be correct. More quantitatively, we can make the

$$\sin(m\phi) \to \sin(m(\phi - 90°/m)) = \sin(m\phi - 90°) = -\cos(m\phi) \tag{27}$$

and

$$\cos(m\phi) \to \cos(m(\phi - 90°/m)) = \cos(m\phi - 90°) = \sin(m\phi) \tag{28}$$

replacements in above equations for plotting the $E_z \propto \sin(m\phi)$ modes properly.

Now, how are these normal mode fields polarized? The answer is, a variety of polarizations will be found, depending on each mode. Later on we will find out that it is possible to organize some groups of the allowed modes in certain linear combinations that yield linearly polarized field distributions in x or y directions if the condition $\delta n = n_1 - n_2 \ll n_1$ holds. But before we study those "quasi LP modes" of "weakly guided fibers" we will take a look at the exact normal modes corresponding to the direct solutions we have derived above — they are called TE and TM modes when $m = 0$, and HE and EH modes when $m \geq 1$.

1.3 TE and TM modes

Let $m = 0$. Then our dispersion relation simplifies as

$$(\frac{J_0'(X)}{XJ_0(X)} + \frac{K_0'(Y)}{YK_0(Y)})(\frac{n_1^2 J_0'(X)}{XJ_0(X)} + \frac{n_2^2 K_0'(Y)}{YK_0(Y)}) = 0 \tag{29}$$

while the fields in the core region are given, when $E_z \propto \cos(m\phi) = 1$ and $H_z = 0$, by TM mode equations

$$\begin{aligned}
E_\phi &= \frac{-j}{k_t^2}(\frac{-m\beta}{r}AJ_m(k_tr)\sin(m\phi) - \omega\mu_1 Bk_tJ_m'(k_tr)\sin(m\phi))e^{-j\beta z} = 0 \\
E_r &= \frac{-j}{k_t^2}(\frac{m\omega\mu}{r}BJ_m(k_tr)\cos(m\phi) + \beta k_tAJ_m'(k_tr)\cos(m\phi))e^{-j\beta z} = \frac{-j}{k_t^2}\beta k_tAJ_m'(k_tr)e^{-j\beta z} \\
H_\phi &= \frac{-j}{k_t^2}(\omega\epsilon_1 k_tAJ_m'(k_tr)\cos(m\phi) + \frac{m\beta}{r}BJ_m(k_tr)\cos(m\phi))e^{-j\beta z} = \frac{-j}{k_t^2}\omega\epsilon_1 k_tAJ_m'(k_tr)e^{-j\beta z} \\
H_r &= \frac{-j}{k_t^2}(\beta k_tBJ_m'(k_tr)\sin(m\phi) + \frac{m\omega\epsilon}{r}AJ_m(k_tr)\sin(m\phi))e^{-j\beta z} = 0.
\end{aligned} \tag{30}$$

With $E_z \propto \sin(m\phi) = 0$ and $H_z \propto \cos(m\phi) = 1$, we have the TE mode equations

$$\begin{aligned}
E_\phi &= \frac{-j}{k_t^2}(\frac{m\beta}{r}AJ_m(k_tr)\cos(m\phi) + \omega\mu_1 Bk_tJ_m'(k_tr)\cos(m\phi))e^{-j\beta z} = \frac{-j}{k_t^2}\omega\mu_1 Bk_tJ_m'(k_tr)e^{-j\beta z} \\
E_r &= \frac{-j}{k_t^2}(\frac{m\omega\mu}{r}BJ_m(k_tr)\sin(m\phi) + \beta k_tAJ_m'(k_tr)\sin(m\phi))e^{-j\beta z} = 0 \\
H_\phi &= \frac{-j}{k_t^2}(\omega\epsilon_1 k_tAJ_m'(k_tr)\sin(m\phi) + \frac{m\beta}{r}BJ_m(k_tr)\sin(m\phi))e^{-j\beta z} = 0 \\
H_r &= \frac{j}{k_t^2}(\beta k_tBJ_m'(k_tr)\cos(m\phi) + \frac{m\omega\epsilon}{r}AJ_m(k_tr)\cos(m\phi))e^{-j\beta z} = \frac{j}{k_t^2}\beta k_tBJ_m'(k_tr)e^{-j\beta z}.
\end{aligned} \tag{31}$$

Note that TE mode is characterized by an azimuthal polarized electric field while the TM mode is carried by a radially polarized field. One of them, TE mode, has the dispersion relation

$$\frac{J_0'(X)}{XJ_0(X)} = -\frac{K_0'(Y)}{YK_0(Y)} \tag{32}$$

while the other, TM mode,

$$\frac{n_1^2}{n_2^2}\frac{J_0'(X)}{XJ_0(X)} = -\frac{K_0'(Y)}{YK_0(Y)}). \tag{33}$$

This identification of the two separate roots of (24) for $m = 0$ case with TE and TM modes comes from comparisons of (32) and (33) with (22) and (23), respectively. (22) came from E_ϕ matching when $H_z \neq 0$, meaning TE when $m = 0$. Likewise, (23) came from H_ϕ matching when $E_z \neq 0$, meaning TM when $m = 0$.

The distinctions between TE mode dispersion relation (32) and TM wave relation (33) are negligible when $n_1 - n_2 \ll n_1$. In that case (32) applies equally well for both TE and TM modes.

Recall that

$$V = k_o a\sqrt{n_1^2 - n_2^2} = \frac{2\pi a}{\lambda_o}\sqrt{n_1^2 - n_2^2} = \frac{\omega}{c}a\sqrt{n_1^2 - n_2^2}.$$

Using this expression with $\lambda_o = 1.5\,\mu$m, $a = 5\lambda_o$, $n_1 = 1.5$, $\delta n = n_1 - n_2 = 0.02$ we find that $V \approx 7.7$. If the fiber radius is reduced to $a = \lambda_o$ then V-number is reduced to about 1.53.

In Figure 2 we depict a plot of both sides of (32) versus X when $V = 2$ and 8. The RHS curves in such figures occupy the range $0 < X < V$ as explained below:

```
V1 = 2; V2 = 8; Y1 = √(V1² - X²) ; Y2 = √(V2² - X²) ; m = 0;
Jm = BesselJ[m, X]; Jmp = -BesselJ[m + 1, X] + m * BesselJ[m, X] / X;
Km1 = BesselK[m, Y1]; Kmp1 = (-1)^1 * BesselK[m + 1, Y1] + m * BesselK[m, Y1] / Y1;
Km2 = BesselK[m, Y2]; Kmp2 = (-1)^1 * BesselK[m + 1, Y2] + m * BesselK[m, Y2] / Y2;
LHS = Jmp / (X * Jm); RHS1 = - Kmp1 / (Y1 * Km1); RHS2 = - Kmp2 / (Y2 * Km2);
Show[{Plot[{LHS, RHS1, RHS2}, {X, 0, 10}, AxesLabel → {"X", None}],
  Graphics[{Line[{{V1, -1.5}, {V1, 1.5}}], Text["V=" + V1, {V1, -1.2}],
   Line[{{V2, -1.5}, {V2, 1.5}}], Text["V=" + V2, {V2, -1.2}],
   Text["RHS", {0.5, 0.35}], Text["LHS", {0.5, -0.7}]}]}]
(* m=0 --- TE and TM modes --- TE/TM01 and 02 are excited for V=8 *)
```



Figure 2: TE0l and TM0l modes obtained with $m = 0$.

Notice that with $V = 8$, the intersections of the RHS and LHS curves in the range $X < V = 8$ imply the propagation of TE01, TM01, TE02, and TM02 modes in a fiber with a *V-number* of 8. But when the *V-number* is reduced below 2.405 all of those modes would be put into "cutoff" — this is clear from the $V = 2$ curve included in the plot. When we examine the HE and EH modes in the next section, we will find out the existence of a "fundamental" HE11 mode that enjoys the fiber all by itself if $V < 2.405$. This critical number, $V_c = 2.405$, is *the first zero of* $J_0(X)$ in X associated with the "first infinities" of LHS curve $\propto 1/J_0(X)$!! In fact all the "infinities" of the LHS curve shown (in blue) in Figure 2 correspond to successive zeroes of $J_0(X)$, namely, 2.405, 5.520, 8.655, 11.792, etc. (verify in HW).

In examining Figure 2, and the subsequent dispersion plots to be shown, take a notice that the oscillatory and blue colored LHS curves depend only on X — they are, most importantly, V-invariant! The RHS curves, on the other hand, are very sensitive to V, and in fact only exist as *real-valued functions* for $X < V$. Thus "guided solutions" on optical fibers are only possible for $X < V$ (otherwise Y is imaginary and no good), and the existence of propagating modes in this permissible range of X depends on whether the range is populated by intersecting LHS curves for possible modes — there is no such LHS curve in Figure 2 for $X < 2.405$.

1.4 HE and EH modes

Let's take a look at the dispersion relation (24) once more which is repeated here:

$$\left(\frac{J'_m(X)}{XJ_m(X)} + \frac{K'_m(Y)}{YK_m(Y)}\right)\left(\frac{n_1^2 J'_m(X)}{XJ_m(X)} + \frac{n_2^2 K'_m(Y)}{YK_m(Y)}\right) = \frac{m^2\beta^2}{k_o^2}\left(\frac{1}{X^2} + \frac{1}{Y^2}\right)^2.$$

We will solve this dispersion relation when $m \geq 1$ after making some simplifying approximations. Let $n_1^2 \approx n_2^2 \approx n^2$ in which case we have

$$(\frac{J'_m(X)}{XJ_m(X)} + \frac{K'_m(Y)}{YK_m(Y)})^2 \approx \frac{m^2\beta^2}{n^2k_o^2}(\frac{1}{X^2} + \frac{1}{Y^2})^2 \approx m^2(\frac{1}{X^2} + \frac{1}{Y^2})^2$$

since if $n_1 \approx n_2 \approx n$ then $\beta \approx nk_o$ — essentially the case of "weakly guided" waves with negligible k_t and α. It then follows that

$$\frac{J'_m(X)}{XJ_m(X)} + \frac{K'_m(Y)}{YK_m(Y)} \approx \pm m(\frac{1}{X^2} + \frac{1}{Y^2})$$

leading to

$$\frac{J'_m(X)}{XJ_m(X)} \approx -\frac{K'_m(Y)}{YK_m(Y)} \pm m(\frac{1}{X^2} + \frac{1}{Y^2}). \tag{34}$$

These two approximate roots of our exact dispersion relation (24) for $m \geq 1$ are the weakly guided HE (use $-m$; E_z dominates H_z in this mode) and EH (use $+m$; H_z dominates E_z in this mode) modes as we will find out. Also, with $m = 0$ this single dispersion relation covers the weakly guided TE and TM modes that we have already looked at[2].

Obtaining the exact roots of (24) for $m \geq 1$, in order to study the "strongly guided" EH and HE modes, is quite cumbersome and will not be attempted here — if interested in doing that, use $\beta^2 = k_o^2 n_1^2 - X^2/a^2$ in the dispersion equation (24) and use a numerical root finder. Fortunately doing that is not necessary because in practical optical fibers utilized in the low-loss $\lambda_o \approx 1.5\,\mu$m band (attenuation dips down to about 0.15 dB/km around here — see Figure 3) we have $n_1 \approx n_2 \approx 1.5$ (fine tuned by doping *fused silica* SiO_2 with, e.g., GeO_2 to increase its n within the core) and the weak guiding approximation works very well.



Figure 9.3-2 Attenuation coefficient α of silica glass versus wavelength λ_o. There is a local minimum at 1.3 μm ($\alpha \approx 0.3$ dB/km) and an absolute minimum at 1.55 μm ($\alpha \approx 0.15$ dB/km).

Figure 3: The attenuation constant of fused silica within the 1 μm wavelength band — from *Saleh and Teisch* (2007).

[2]Notice that TM0l and TE0l modes correspond to $\cos(m\phi)$ and $\sin(m\phi)$ normal modes of E_z, respectively, when $m = 0$. With $m \geq 1$, each EHml and HEml mode (selected by $\pm m$ choice in (34)) will have each of these $\cos(m\phi)$ and $\sin(m\phi)$ normal modes of E_z. Because of this we say that EH and HE modes have a "2-fold degeneracy" while TE and TM modes are not degenerate.

```
V1 = 2; V2 = 8; Y1 = Sqrt[V1^2 - X^2]; Y2 = Sqrt[V2^2 - X^2]; m = 1;
Jm = BesselJ[m, X]; Jmp = -BesselJ[m + 1, X] + m * BesselJ[m, X] / X;
Km1 = BesselK[m, Y1]; Kmp1 = (-1)^1 * BesselK[m + 1, Y1] + m * BesselK[m, Y1] / Y1;
Km2 = BesselK[m, Y2]; Kmp2 = (-1)^1 * BesselK[m + 1, Y2] + m * BesselK[m, Y2] / Y2;
LHS = Jmp/(X * Jm); RHS1 = -Kmp1/(Y1 * Km1) - m (1/X^2 + 1/Y1^2); RHS2 = -Kmp2/(Y2 * Km2) - m (1/X^2 + 1/Y2^2);
Show[{Plot[{LHS, RHS1, RHS2}, {X, 0, 10}, AxesLabel → {"X", None}],
  Graphics[{Line[{{V1, -1.5}, {V1, 1.5}}], Text["V=" + V1, {V1, -1.2}],
    Line[{{V2, -1.5}, {V2, 1.5}}], Text["V=" + V2, {V2, -1.2}],
    Text["RHS", {0.5, -0.8}], Text["LHS", {0.5, 0.8}]}]}]
(* m=1 --- HE modes --- only HE11 if V=2 + HE12 & HE13 if V=8 *)
```

(a) 


```
V1 = 2; V2 = 8; Y1 = Sqrt[V1^2 - X^2]; Y2 = Sqrt[V2^2 - X^2]; m = 1;
Jm = BesselJ[m, X]; Jmp = -BesselJ[m + 1, X] + m * BesselJ[m, X] / X;
Km1 = BesselK[m, Y1]; Kmp1 = (-1)^1 * BesselK[m + 1, Y1] + m * BesselK[m, Y1] / Y1;
Km2 = BesselK[m, Y2]; Kmp2 = (-1)^1 * BesselK[m + 1, Y2] + m * BesselK[m, Y2] / Y2;
LHS = Jmp/(X * Jm); RHS1 = -Kmp1/(Y1 * Km1) + m (1/X^2 + 1/Y1^2); RHS2 = -Kmp2/(Y2 * Km2) + m (1/X^2 + 1/Y2^2);
Show[{Plot[{LHS, RHS1, RHS2}, {X, 0, 10}, AxesLabel → {"X", None}],
  Graphics[{Line[{{V1, -4.5}, {V1, 4.5}}], Text["V=" + V1, {V1, -1.2}],
    Line[{{V2, -4.5}, {V2, 4.5}}], Text["V=" + V2, {V2, -1.2}],
    Text["RHS", {1.5, 1.5}], Text["LHS", {0.5, 0.8}]}]}]
(* m=1 --- EH modes --- no modes if V=2 but EH11 & EH12 if V=8 *)
```

(b) 

Figure 4: (a) HE1l modes, (b)EH1l modes obtained with $m = 1$.

Figures 4a and b show the HE and EH mode hybrid solutions for the same *weak-guiding* optical fiber examined in Figure 2 — i.e., $V = 2$ and 8. Also Figure 5 shows the case $m = 2$ for HE modes.

Clearly, an examination of Figures 4a and b and comparisons with Figure 2 show that the *fundamental mode* in a weakly guided fiber is HE11 — it has no cutoff frequency, it propagates at all frequencies ω, or with all *normalized frequencies* V!

Furthermore, designing a fiber with a *V-number* of 2.405 would put all the other modes into cutoff and ensure a "single mode operation". This can be confirmed by an examination of Figure 5 where for $V = 2$ the HE21 mode is cut-off, but it is propagating for $V = 8 > 2.405 \equiv V_{cLP11}$.

```
V1 = 2; V2 = 8; Y1 = Sqrt[V1^2 - X^2]; Y2 = Sqrt[V2^2 - X^2]; m = 2;
Jm = BesselJ[m, X]; Jmp = -BesselJ[m + 1, X] + m * BesselJ[m, X] / X;
Km1 = BesselK[m, Y1]; Kmp1 = (-1)^1 * BesselK[m + 1, Y1] + m * BesselK[m, Y1] / Y1;
Km2 = BesselK[m, Y2]; Kmp2 = (-1)^1 * BesselK[m + 1, Y2] + m * BesselK[m, Y2] / Y2;
LHS = Jmp/(X * Jm); RHS1 = -Kmp1/(Y1 * Km1) - m (1/X^2 + 1/Y1^2); RHS2 = -Kmp2/(Y2 * Km2) - m (1/X^2 + 1/Y2^2);
Show[{Plot[{LHS, RHS1, RHS2}, {X, 0, 10}, AxesLabel → {"X", None}],
  Graphics[{Line[{{V1, -1.5}, {V1, 1.5}}], Text["V=" + V1, {V1, -1.2}],
    Line[{{V2, -1.5}, {V2, 1.5}}], Text["V=" + V2, {V2, -1.2}],
    Text["RHS", {0.5, -0.8}], Text["LHS", {0.5, 0.8}]}]}]
(* m=2 --- HE modes --- no modes if V=2 but HE21 & HE22 if V=8 *)
```



Figure 5: HE2l modes obtained with $m = 2$.

To summarize, so-called "single-mode" fibers have *V-numbers* of 2.405 or less — this critical V-value is independent of fiber constitution and it comes from boundary condition math in cylindrical settings, namely the first zero of $J_0(X)$, much like $\pi/2$ is the first zero of $\cos(X)$. To maintain single mode operation we select a/λ_o, n_1, and δn such that $V \leq 2.405$.

1.5 Field structure of HE11 mode

There are two flavors of the dominant HE11 mode — $\cos(\phi)$ and $\sin(\phi)$ flavors — each having its own preferred polarization direction that is *linear*. It turns out that $\cos(\phi)$ flavor is x-polarized and $\sin(\phi)$ flavor is y-polarized. This makes sense because as x-polarized transverse field loops back into itself via E_z, the latter will need to switch sign as it crosses the $x = 0$ plane just like the $\cos(\phi)$ function. In Figure 6 we show the Mathematica snippet to compute and display vector plots of both flavors of the HE11-mode electric fields in the transverse xy-plane along with the plots themselves — $V = 3$ situation is depicted in the figure using the weak guiding solutions.

```
V = 3; Y = Sqrt[V^2 - X^2]; m = 1; (* solves and plots HE11 mode fields *)
Jm = BesselJ[m, X]; Jmp = -BesselJ[m + 1, X] + m * BesselJ[m, X] / X;
Km = BesselK[m, Y]; Kmp = (-1)^1 * BesselK[m + 1, Y] + m * BesselK[m, Y] / Y;
LHS = Jmp/(X * Jm); RHS = -Kmp/(Y * Km) - m (1/X^2 + 1/Y^2);
p0 = Plot[{LHS, RHS}, {X, 0, 8}, AxesLabel → {"X", None}, ImageSize → {200, 200}];
Xs = X /. NSolve[LHS == RHS, X, Reals][[2, 1]]; Ys = Sqrt[V^2 - Xs^2];
{Xs, Ys, V}
r = Sqrt[x^2 + y^2]; p = ArcTan[x, y];
Jms = BesselJ[m, Xs]; Jmps = -BesselJ[m + 1, Xs] + m * BesselJ[m, Xs] / (Xs);
Kms = BesselK[m, Ys]; Kmps = (-1)^1 * BesselK[m + 1, Ys] + m * BesselK[m, Ys] / (Ys);
Ap = 1; Bp = (-(1/Xs^2 + 1/Ys^2) * m)/(Jmps/(Xs*Jms) + Kmps/(Ys*Kms));
Jmsr = BesselJ[m, Xs * r]; Jmpsr = -BesselJ[m + 1, Xs * r] + m * BesselJ[m, Xs * r] / (Xs * r);
Kmsr = BesselK[m, Ys * r]; Kmpsr = (-1)^1 * BesselK[m + 1, Ys * r] + m * BesselK[m, Ys * r] / (Ys * r);
Cm = Cos[m * p]; Sm = Sin[m * p];

Ez = Jmsr * Cm;
Ep11x = (-m * Ap * Jmsr * Sm / r - Bp * Xs * Jmpsr * Sm);
Er11x = (m * Bp * Jmsr * Cm / r + Xs * Ap * Jmpsr * Cm);
p1 = VectorPlot[{Er11x * Cos[p] - Ep11x * Sin[p], Er11x * Sin[p] + Ep11x * Cos[p]},
    {x, -1, 1}, {y, -1, 1}, PlotLabel → "HE11x", ImageSize → {200, 200}];

Ez = Jmsr * Sm;
Ep11y = (m * Ap * Jmsr * Cm / r + Bp * Xs * Jmpsr * Cm);
Er11y = (m * Bp * Jmsr * Sm / r + Xs * Ap * Jmpsr * Sm);
p2 = VectorPlot[{Er11y * Cos[p] - Ep11y * Sin[p], Er11y * Sin[p] + Ep11y * Cos[p]},
    {x, -1, 1}, {y, -1, 1}, PlotLabel → "HE11y", ImageSize → {200, 200}];
GraphicsGrid[{{p0, p1, p2}}]
{1.77109, 2.42141, 3}
```



Figure 6: Mathematica code and plots of transverse electric fields for both flavors of the fundamental or *dominant* HE11 mode.

```
V = 3; Y = Sqrt[V^2 - X^2]; m = 0; (* solves and plots TM01 and TE01 mode fields *)
Jm = BesselJ[m, X]; Jmp = -BesselJ[m + 1, X] + m * BesselJ[m, X] / X;
Km = BesselK[m, Y]; Kmp = (-1)^1 * BesselK[m + 1, Y] + m * BesselK[m, Y] / Y;
LHS = Jmp / (X * Jm); RHS = - Kmp / (Y * Km);
p0 = Plot[{LHS, RHS}, {X, 0, 8}, AxesLabel → {"X", None}, ImageSize → {200, 200}];
Xs = X /. NSolve[LHS == RHS, X, Reals][[2, 1]]; Ys = Sqrt[V^2 - Xs^2];
{Xs, Ys, V}
r = Sqrt[x^2 + y^2]; p = ArcTan[x, y];
Jmsr = BesselJ[m, Xs * r]; Jmpsr = -BesselJ[m + 1, Xs * r] + m * BesselJ[m, Xs * r] / (Xs * r);
Cm = Cos[m * p]; Sm = Sin[m * p];
Ez01 = Jmsr * Cm;
Er01 = -Jmpsr;
Ep01 = -Jmpsr;
p1 = VectorPlot[{Er01 * Cos[p], Er01 * Sin[p]},
    {x, -1, 1}, {y, -1, 1}, PlotLabel → "TM01", ImageSize → {200, 200}];
p2 = VectorPlot[{-Ep01 * Sin[p], Ep01 * Cos[p]}, {x, -1, 1},
    {y, -1, 1}, PlotLabel → "TE01", ImageSize → {200, 200}];
GraphicsGrid[{{p0, p1, p2}}]
{2.71907, 1.26754, 3}
```



Figure 7: Mathematica code and plots of transverse electric fields for radial TM01 and azimuthal TE01 modes.

1.6 LP quasi-modes

Neither TE and TM nor HEml and EHml modes for $m > 1$ are linearly polarized — Figures 7 and 8 show the transverse field distributions of TM01/TE01 and HE21 modes as useful examples.

However, it does turn out that for weakly guided fibers one could construct linearly polarized superpositions of TM01, TE01, and HE21 modes — these are classified as a four-fold degenerate "LP11 mode". This is possible because under the weak guiding condition propagation constants β and cutoff frequencies V_c of these three modes are nearly identical.

```
V = 3; Y = Sqrt[V^2 - X^2]; m = 2; (* solves and plots HE21 mode fields *)
Jm = BesselJ[m, X]; Jmp = -BesselJ[m + 1, X] + m * BesselJ[m, X] / X;
Km = BesselK[m, Y]; Kmp = (-1)^1 * BesselK[m + 1, Y] + m * BesselK[m, Y] / Y;
LHS = Jmp/(X * Jm); RHS = -Kmp/(Y * Km) - m (1/X^2 + 1/Y^2);
p0 = Plot[{LHS, RHS}, {X, 0, 8}, AxesLabel → {"X", None}, ImageSize → {200, 200}];
Xs = X /. NSolve[LHS == RHS, X, Reals][[2, 1]]; Ys = Sqrt[V^2 - Xs^2];
{Xs, Ys, V}
r = Sqrt[x^2 + y^2]; p = ArcTan[x, y];
Jms = BesselJ[m, Xs]; Jmps = -BesselJ[m + 1, Xs] + m * BesselJ[m, Xs] / (Xs);
Kms = BesselK[m, Ys]; Kmps = (-1)^1 * BesselK[m + 1, Ys] + m * BesselK[m, Ys] / (Ys);
Ap = 1; Bp = (-(1/Xs^2 + 1/Ys^2) * m) / (Jmps/(Xs*Jms) + Kmps/(Ys*Kms));
Jmsr = BesselJ[m, Xs * r]; Jmpsr = -BesselJ[m + 1, Xs * r] + m * BesselJ[m, Xs * r] / (Xs * r);
Kmsr = BesselK[m, Ys * r]; Kmpsr = (-1)^1 * BesselK[m + 1, Ys * r] + m * BesselK[m, Ys * r] / (Ys * r);
Cm = Cos[m * p]; Sm = Sin[m * p];

Ez = Jmsr * Cm;
Ep21x = (-m * Ap * Jmsr * Sm / r - Bp * Xs * Jmpsr * Sm);
Er21x = (m * Bp * Jmsr * Cm / r + Xs * Ap * Jmpsr * Cm);
p1 = VectorPlot[{Er21x * Cos[p] - Ep21x * Sin[p], Er21x * Sin[p] + Ep21x * Cos[p]},
    {x, -1, 1}, {y, -1, 1}, PlotLabel → "HE21x", ImageSize → {200, 200}];

Ez = Jmsr * Sm;
Ep21y = (m * Ap * Jmsr * Cm / r + Bp * Xs * Jmpsr * Cm);
Er21y = (m * Bp * Jmsr * Sm / r + Xs * Ap * Jmpsr * Sm);
p2 = VectorPlot[{Er21y * Cos[p] - Ep21y * Sin[p], Er21y * Sin[p] + Ep21y * Cos[p]},
    {x, -1, 1}, {y, -1, 1}, PlotLabel → "HE21y", ImageSize → {200, 200}];
GraphicsGrid[{{p0, p1, p2}}]
{2.71907, 1.26754, 3}
```



Figure 8: Mathematica code and plots of transverse electric fields for both flavors of HE21 mode.

Figures 9-10 illustrate how various "weighted" superpositions of TM01, TE01, and HE21 modes produce all four flavors of the linearly polarized LP11 mode.

```
Er = Er01; Ep = 0;
p1 = VectorPlot[{Er * Cos[p] - Ep * Sin[p], Er * Sin[p] + Ep * Cos[p]},
    {x, -1, 1}, {y, -1, 1}, PlotLabel → "TM01", ImageSize → {200, 200}];
Er = Er21x; Ep = Ep21x;
p2 = VectorPlot[{Er * Cos[p] - Ep * Sin[p], Er * Sin[p] + Ep * Cos[p]},
    {x, -1, 1}, {y, -1, 1}, PlotLabel → "HE21x", ImageSize → {200, 200}];
Er = 2.7 * Er01 + Er21x; Ep = 0 + Ep21x;
p3 = VectorPlot[{Er * Cos[p] - Ep * Sin[p], Er * Sin[p] + Ep * Cos[p]},
    {x, -1, 1}, {y, -1, 1}, PlotLabel → "TM01+HE21x", ImageSize → {200, 200}];
GraphicsGrid[{{p1, p2, p3}}]
```



Figure 9: LP11xx

```
Er = 0; Ep = Ep01;
p1 = VectorPlot[{Er * Cos[p] - Ep * Sin[p], Er * Sin[p] + Ep * Cos[p]},
    {x, -1, 1}, {y, -1, 1}, PlotLabel → "TE01", ImageSize → {200, 200}];
Er = Er21y; Ep = Ep21y;
p2 = VectorPlot[{Er * Cos[p] - Ep * Sin[p], Er * Sin[p] + Ep * Cos[p]},
    {x, -1, 1}, {y, -1, 1}, PlotLabel → "HE21y", ImageSize → {200, 200}];
Er = Er21y - 0; Ep = Ep21y - 2.7 * Ep01;
p3 = VectorPlot[{Er * Cos[p] - Ep * Sin[p], Er * Sin[p] + Ep * Cos[p]},
    {x, -1, 1}, {y, -1, 1}, PlotLabel → "HE21y-TE01", ImageSize → {200, 200}];
GraphicsGrid[{{p1, p2, p3}}]
```



(a)

```
Er = 0; Ep = Ep01;
p1 = VectorPlot[{Er * Cos[p] - Ep * Sin[p], Er * Sin[p] + Ep * Cos[p]},
    {x, -1, 1}, {y, -1, 1}, PlotLabel → "TE01", ImageSize → {200, 200}];
Er = Er21y; Ep = Ep21y;
p2 = VectorPlot[{Er * Cos[p] - Ep * Sin[p], Er * Sin[p] + Ep * Cos[p]},
    {x, -1, 1}, {y, -1, 1}, PlotLabel → "HE21y", ImageSize → {200, 200}];
Er = Er21y + 0; Ep = Ep21y + 2.7 * Ep01;
p3 = VectorPlot[{Er * Cos[p] - Ep * Sin[p], Er * Sin[p] + Ep * Cos[p]},
    {x, -1, 1}, {y, -1, 1}, PlotLabel → "HE21y+TE01", ImageSize → {200, 200}];
GraphicsGrid[{{p1, p2, p3}}]
```



(b)

```
Er = Er01; Ep = 0;
p1 = VectorPlot[{Er * Cos[p] - Ep * Sin[p], Er * Sin[p] + Ep * Cos[p]},
    {x, -1, 1}, {y, -1, 1}, PlotLabel → "TM01", ImageSize → {200, 200}];
Er = Er21x; Ep = Ep21x;
p2 = VectorPlot[{Er * Cos[p] - Ep * Sin[p], Er * Sin[p] + Ep * Cos[p]},
    {x, -1, 1}, {y, -1, 1}, PlotLabel → "HE21x", ImageSize → {200, 200}];
Er = 2.7 * Er01 - Er21x; Ep = 0 - Ep21x;
p3 = VectorPlot[{Er * Cos[p] - Ep * Sin[p], Er * Sin[p] + Ep * Cos[p]},
    {x, -1, 1}, {y, -1, 1}, PlotLabel → "TM01-HE21x", ImageSize → {200, 200}];
GraphicsGrid[{{p1, p2, p3}}]
```



(c)

Figure 10: (a) LP11xy, (b) LP11yx, (c) LP11yy

Furthermore similar groupings of higher order TE, TM, HE, and EH modes into higher-order quasi-linear-polarized modes labelled as "LPml modes" turns out to be possible. Some of such groupings of orders $m = 0$ and 1 and their relationships to the zeroes of $J_0(X)$ and $J_1(X)$ functions are illustrated in Figure 11 taken from *Gloge* [1971].



Figure 11: The regions of parameter X for modes of order $m = 0, 1$ — from *Gloge*, Applied Optics, **10**, 1971.

We next furnish a simple derivation of linearly polarized LP-mode fields and their dispersion relations following *Gloge* [1971].

Postulate that transverse components of the vector wave equation describing a step-index fiber can be satisfied by the linear polarized fields

$$E_x = \frac{H_y}{\eta_1} = e^{-j\beta z} \left\{ \begin{array}{c} \cos(m\phi) \\ \sin(m\phi) \end{array} \right\} \frac{J_m(\frac{Xr}{a})}{J_m(X)} \tag{35}$$

and

$$E_x = \frac{H_y}{\eta_2} = e^{-j\beta z} \left\{ \begin{array}{c} \cos(m\phi) \\ \sin(m\phi) \end{array} \right\} \frac{K_m(\frac{Yr}{a})}{K_m(Y)} \tag{36}$$

within core and cladding regions, respectively, while $E_y = H_x = 0$ everywhere. With these assumptions, and focusing only on $\cos(m\phi)$ mode for the moment, we can compute the E_z and H_z from Maxwell's curl equations as

$$E_z = \frac{1}{j\omega\epsilon}\frac{\partial H_y}{\partial x} \text{ and } H_z = -\frac{1}{j\omega\mu}\frac{\partial E_x}{\partial y} \tag{37}$$

in the absence of H_x and E_y (HW problem). We start with

$$E_z = \frac{1}{j\omega\epsilon}\frac{\partial H_y}{\partial x} = e^{-j\beta z} \left\{ \begin{array}{c} \frac{\eta_1}{j\omega\epsilon_1}[\cos(m\phi)\frac{XJ'_m(\frac{Xr}{a})}{aJ_m(X)}\frac{\partial r}{\partial x} + \frac{J_m(\frac{Xr}{a})}{J_m(X)}\frac{\partial}{\partial x}\cos(m\phi)] \\ \frac{\eta_2}{j\omega\epsilon_2}[\cos(m\phi)\frac{YK'_m(\frac{Yr}{a})}{aK_m(Y)}\frac{\partial r}{\partial x} + \frac{K_m(\frac{Yr}{a})}{K_m(Y)}\frac{\partial}{\partial x}\cos(m\phi)] \end{array} \right\} \tag{38}$$

inside and outside the core, and, likewise

$$H_z = \frac{-1}{j\omega\mu}\frac{\partial E_x}{\partial y} = e^{-j\beta z} \left\{ \begin{array}{c} \frac{-1}{j\omega\mu_1}[\cos(m\phi)\frac{XJ'_m(\frac{Xr}{a})}{aJ_m(X)}\frac{\partial r}{\partial y} + \frac{J_m(\frac{Xr}{a})}{J_m(X)}\frac{\partial}{\partial y}\cos(m\phi)] \\ \frac{-1}{j\omega\mu_2}[\cos(m\phi)\frac{YK'_m(\frac{Yr}{a})}{aK_m(Y)}\frac{\partial r}{\partial y} + \frac{K_m(\frac{Yr}{a})}{K_m(Y)}\frac{\partial}{\partial y}\cos(m\phi)] \end{array} \right\}. \tag{39}$$

We can test the validity of these results by trying to recover (35)-(36) from (38)-(39) by using Maxwell's curl equations once more and finishing the entire algebra started above. When that is done one finds E_x

and H_y expressions a little different from (35)-(36) that we started with, but the differences are negligible when "weak guiding" condition is valid, that is when $\eta_1 \approx \eta_2$, $\mu_1 \approx \mu_2$, $\epsilon_1 \approx \epsilon_2$.

Accordingly, we can accept our postulated and linear polarized solutions (35)-(36) as valid solutions (or good approximations) of the step-index optical fiber problem under weak guiding conditions so long as we identify the "quantized" and permissible *X-numbers* of these LPml modes — that requires matching the tangential field components (38) and (39) at the $r = a$ interface. With $\eta_1 \approx \eta_2$, $\mu_1 \approx \mu_2$, $\epsilon_1 \approx \epsilon_2$ simplifications we obtain from each one of those expressions similar constraints like

$$\cos(m\phi)\frac{XJ'_m(X)}{aJ_m(X)}\frac{\partial r}{\partial x} + \frac{J_m(X)}{J_m(X)}\frac{\partial}{\partial x}\cos(m\phi) = \cos(m\phi)\frac{YK'_m(Y)}{aK_m(Y)}\frac{\partial r}{\partial x} + \frac{K_m(Y)}{K_m(Y)}\frac{\partial}{\partial x}\cos(m\phi)$$

both of which simplify identically as our LPml-mode dispersion relation

$$\frac{XJ'_m(X)}{J_m(X)} = \frac{YK'_m(Y)}{K_m(Y)} \tag{40}$$

after dropping identical terms from both sides of the equality.

We can now obtain the β's of LPml modes from this simple dispersion relation using straightforward graphical procedures that we are very familiar with. Using Bessel identities, this dispersion relation can also be expressed as (HW problem)

$$\frac{XJ_{m\pm1}(X)}{J_m(X)} = \pm\frac{YK_{m\pm1}(Y)}{K_m(Y)}. \tag{41}$$

Note that when using (40) or (41) to compute *X-numbers* of LPml modes from LHS and RHS intersections, remember to start with $m = 0$ in order to identify the *fundamental* mode LP01 that is the same as our good old HE11-mode that was obtained from HEml-mode dispersion relation (34) by using $m = 1$! — so, m's are a little different in these two contexts, be careful!

You can also use (40) or (41) with $m = 1$ to compute the *X-number* of LP11-mode approximating the almost identical *X-numbers of* the TM01, TE01, and HE21 modes that we encountered and studied earlier.

Overall, dispersion relation (41) describes *4-fold degenerate* and weakly-guided LPml modes with transverse core electric fields

$$\mathbf{E}_\perp = e^{-j\beta z}\frac{J_m(\frac{Xr}{a})}{J_m(X)}\left\{\begin{array}{c}\cos(m\phi)\\ \sin(m\phi)\end{array}\right\}\{\hat{x},\hat{y}\} \tag{42}$$

and transverse core magnetic fields

$$\mathbf{H}_\perp = e^{-j\beta z}\frac{J_m(\frac{Xr}{a})}{\eta_1 J_m(X)}\left\{\begin{array}{c}\cos(m\phi)\\ \sin(m\phi)\end{array}\right\}\{\hat{y},-\hat{x}\} \tag{43}$$

accompanying some weak E_z and H_z fields in the propagation direction. These modes for $m \geq 0$ and $l \geq 1$ provide a complete description of the available guided modes in practical step-index fibers to an excellent approximation. The dispersion curves for these modes are shown in Figure 12.

Well, this is THE END.

If you liked all this and want to learn more and experiment with step-index and graded-index fibers in the lab, then take ECE 465 next fall — it will be a new 4 hr course with only ECE 350 prerequisite that will count as an ECE lab.

Erhan Kudeki, Nov 24, 2012

```
LP[m_, V_, X0_] := Module[{X}, Y = Sqrt[V^2 - X^2];
  Jm = BesselJ[m, X]; Jmp = BesselJ[m - 1, X];
  Km = BesselK[m, Y]; Kmp = BesselK[m - 1, Y]; LHS = X * Jmp / Jm; RHS = - Y * Kmp / Km;
  Xs = X /. FindRoot[LHS == RHS, {X, X0}]; {V, Xs}]

LP01 = {{0, 0}, LP[0, 1, 0.2], LP[0, 1.5, 1], LP[0, 2.1, 2],
  LP[0, 3, 2], LP[0, 4, 2], LP[0, 6, 2], LP[0, 8, 2], LP[0, 10, 2]}

{{0, 0}, {1, 0.979311}, {1.5, 1.31689}, {2.1, 1.56082},
 {3, 1.77109}, {4, 1.9069}, {6, 2.05493}, {8, 2.13458}, {10, 2.18452}}

LP11 = {{2.405, 2.405}, LP[1, 3, 2.6], LP[1, 3.5, 3],
  LP[1, 4, 3], LP[1, 5, 3], LP[1, 6, 3], LP[1, 8, 3], LP[1, 10, 3]}

{{2.405, 2.405}, {3, 2.71907}, {3.5, 2.87777},
 {4, 2.99316}, {5, 3.15273}, {6, 3.25929}, {8, 3.39431}, {10, 3.47699}}

LP21 = {{3.832, 3.832}, LP[2, 4.5, 4], LP[2, 5.5, 5], LP[2, 6, 5], LP[2, 8, 5], LP[2, 10, 5]}

{{3.832, 3.832}, {4.5, 4.06333}, {5.5, 4.26908}, {6, 4.3423}, {8, 4.53832}, {10, 4.65441}}

LP02 = {{3.832, 3.832}, LP[0, 4.2, 4],
  LP[0, 4.7, 4.5], LP[0, 5.5, 5], LP[0, 6, 5], LP[0, 8, 5], LP[0, 10, 5]}

{{3.832, 3.832}, {4.2, 4.12588}, {4.7, 4.33917},
 {5.5, 4.54538}, {6, 4.63577}, {8, 4.8658}, {10, 4.99665}}

LP31 = {{5.136, 5.136}, LP[3, 6, 5], LP[3, 7, 5], LP[3, 8, 5], LP[3, 9, 5], LP[3, 10, 5]}

{{5.136, 5.136}, {6, 5.34985}, {7, 5.50879}, {8, 5.62149}, {9, 5.70679}, {10, 5.77402}}

LP12 = {{5.52, 5.52}, LP[1, 6, 5.6], LP[1, 6.5, 5.6],
  LP[1, 7, 5.6], LP[1, 8, 5.6], LP[1, 9, 5.6], LP[1, 10, 5.6]}

{{5.52, 5.52}, {6, 5.74824}, {6.5, 5.88881},
 {7, 5.99246}, {8, 6.14254}, {9, 6.24949}, {10, 6.33103}}

ListLinePlot[{{{0, 0}, {5, 5}, {10, 10}}, LP01, LP11, LP21, LP02, LP31, LP12},
 PlotRange → {{0, 10}, {0, 7}}, Frame → True, FrameLabel → {"V", "X"},
 PlotStyle → {Dashed, Red, Blue, Green, Magenta, Orange, Yellow}]
```



Figure 12: X solutions of LPml-mode dispersion relation as a function of V for LP01 (red), LP11 (blue), LP21 (green), LP02 (magenta), LP31 (orange), LP12 (yellow) modes. Dispersion curves populate only the $X < V$ region (the bottom triangle underneath the dashed $X = V$ line). Each LPml-mode propagates only when $V > V_{c,ml}$, where $V_{c,ml}$ is lth zero of $J_{m-1}(X)$ in X; also, permissible X for each LPml-mode is $V_{c,ml} < X < X_{m,ml}$, where $X_{m,ml}$ is the zero of $J_m(X)$ in X just above $V_{c,ml}$ — see Figure 11 for a check of this rule for $0l$ and $1l$ modes.